
Incrémentation de boucles et couleurs dans source RPG 
 

 

CMD : INBRPG50               

              CMD        PROMPT(' Traitement des Sources RPG  ')                                                  

/* Bibliothèque                 */                                                                               

              PARM       KWD(BIBL) TYPE(*CHAR) LEN(10) DFT(*LIBL) +                                               

                         PROMPT('Bibliothèque . . . . . . . . .')                                               

/* Fichier                      */                                                                               

              PARM       KWD(FICH) TYPE(*CHAR) LEN(10) DFT(QRPGSRC) + 

                         PROMPT('Fichier  . . . . . . . . . . .')                                               

/* Membre                       */                                                                               

              PARM       KWD(MBR) TYPE(*CHAR) LEN(10) +                                                           

                         PROMPT('Source RPG . . . . . . . . . .')                                               

/* Type de Membre               */                                                                               

              PARM       KWD(TYPE) TYPE(*CHAR) LEN(10) DFT(RPG) +                                                 

                         PROMPT('Type de source . . . . . . . .')                                               

/* Texte du Membre              */                                                                               

              PARM       KWD(TEXTE) TYPE(*CHAR) LEN(50)  +                                                        

                         PROMPT('Texte du source  . . . . . . .') 

 

 

PF : INPRPGSR               

A          R RPG                    TEXT('Incréments de boucles et coul-                                 

A                                      eurs ds source RPGLE')                                               

A            SRCSEQ         6S 2                                                                            

A            SRCDAT         6S 0                                                                            

A            SRCDTA       100 

 

 

CL : INCRPG50               

  FICHIER SOURCE  . . . . .  GBTEST/QCLPSRC 

  MEMBRE  . . . . . . . . .  INCRPG50 

    100 /********************************************************************/                                           

    200 /*                                                                  */                                           

    300 /* Nom du programme . . . . . . . . . . . . :   INCRPG50            */                                           

    400 /* Nom de la bibliothèque . . . . . . . . . :   GBTEST              */                                           

    500 /* Fichier source initial . . . . . . . . . :   QCLPSRC             */                                           

    600 /*                                                                  */                                           

    700 /* Texte  . . : INCREMENTS DE BOUCLE ET COULEURS DANS SOURCES RPG   */                                           

    800 /*                                                                  */                                           

    900 /* Option PDM : ZZ ==>  INBRPG50 &L &F &N &S &X                     */                                           

   1000 /*                                                                  */                                           

   1100 /********************************************************************/                                           

   1200                                                                                                                  

   1300      PGM  PARM(&BIBL &FICH &MBR &RPG &TEXTE)                                                                     

   1400                                                                                                                  

   1500              DCL  VAR(&BIBL) TYPE(*CHAR) LEN(10)           /* Biblio  */                                         

   1600              DCL  VAR(&FICH) TYPE(*CHAR) LEN(10)           /* Fichier */                                         

   1700              DCL  VAR(&MBR) TYPE(*CHAR) LEN(10)            /* Membre  */                                         

   1800              DCL  VAR(&RPG) TYPE(*CHAR) LEN(10)            /* Type    */                                         

   1900              DCL  VAR(&TEXTE) TYPE(*CHAR) LEN(50)          /* Texte   */                                         

   2000              DCL  VAR(&MSGF) TYPE(*CHAR) LEN(70)                                                                 

   2100              DCL  VAR(&NBR) TYPE(*CHAR) LEN(4)                                                                   

   2200                                                                                                                  

   2300  /* Le type de membre doit étre RPG */                                                                           

   2400                                                                                                                  

   2500              IF (&RPG *NE 'RPG') GOTO £ERREUR                                                                    

   2600                                                                                                                  

   2700  /* Le Membre doit exister          */                                                                           

   2800                                                                                                                  

   2900              CHKOBJ     OBJ(&BIBL/&FICH) OBJTYPE(*FILE) MBR(&MBR)                                                

   3000              MONMSG     MSGID(CPF0000)  EXEC(GOTO £ERREUR)                                                       

   3100                                                                                                                  

   3200  /* Création du Source de Travail dans QTEMP */                                                                  

   3300                                                                                                                  

   3400              CHKOBJ     OBJ(QTEMP/INPRPGSR) OBJTYPE(*FILE)                                                       

   3500              MONMSG     MSGID(CPF9801) EXEC(GOTO £CREAT)                                                         

   3600              GOTO       £CLEAR                                                                                   

   3700                                                                                                                  

   3800  £CREAT:     CRTDUPOBJ  OBJ(INPRPGSR) FROMLIB(ZSYS) +                                                            

   3900                           OBJTYPE(*FILE) TOLIB(QTEMP)                                                            

   4000  £CLEAR:     CLRPFM     FILE(QTEMP/INPRPGSR)                                                                     

   4100                                                                                                                  

   4200  /* Copy du Membre origine dans Travail de QTEMP */                                                              

   4300                                                                                                                  

   4400              CPYSRCF    FROMFILE(&BIBL/&FICH) TOFILE(QTEMP/INPRPGSR) +                                           

   4500                           FROMMBR(&MBR) TOMBR(*FIRST)                                                            

   4600                                                                                                                  

   4700  /* Traitement en cours                          */                                                              

   4800                                                                                                                  

   4900              CHGVAR     VAR(&MSGF) VALUE(&MBR *CAT &BIBL *CAT &TEXTE)                                            

   5000              SNDPGMMSG  MSGID(SYS0001) MSGF(GBTEST/INERMSGF) +                                                  

   5100                           MSGDTA(&MSGF) TOPGMQ(*EXT) +                                                           

   5200                           TOMSGQ(*TOPGMQ) MSGTYPE(*STATUS)                                                       

   5300                                                                                                                  

   5400              OVRDBF     FILE(INPRPGSR) TOFILE(QTEMP/INPRPGSR)                                                    

   5500              CALL       PGM(INRRPG50) PARM(&NBR)                                                                 

   5600              DLTOVR     FILE(*ALL)                                                                               

   5700                                                                                                                  

   5800  /* Remplacement du Membre origine               */                                                              

   5900                                                                                                                  

   6000              CPYSRCF    FROMFILE(QTEMP/INPRPGSR) TOFILE(&BIBL/&FICH) +                                           

   6100                           FROMMBR(*FIRST) TOMBR(&MBR)                                                            

   6200                                                                                                                  

   6300              CHGVAR     VAR(&MSGF) VALUE(&NBR *CAT &MBR *CAT &FICH *CAT &BIBL)                                   

   6400              SNDPGMMSG  MSGID(SYS0004) MSGF(GBTEST/INERMSGF) +                                                  

   6500                           MSGDTA(&MSGF) TOPGMQ(*PRV) +                                                           

   6600                           TOMSGQ(*TOPGMQ) MSGTYPE(*COMP)                                                         

   6700              GOTO       £FIN                                                                                     

   6800                                                                                                                  

   6900 £ERREUR:   /* Membre non RPG ou non Trouvé  */                                                                   

   7000                                                                                                                  

   7100              CHGVAR     VAR(&MSGF) VALUE(&MBR *CAT &FICH *CAT &BIBL)                                             

   7200              SNDPGMMSG  MSGID(SYS0002) MSGF(GBTEST/INERMSGF) +                                                  

   7300                           MSGDTA(&MSGF) TOPGMQ(*PRV) +                                                           

   7400                           TOMSGQ(*TOPGMQ) MSGTYPE(*COMP)                                                         

   7500                                                                                                                  

   7600  £FIN:       RCLRSC                                                                                              

   7700      ENDPGM                                                                           

 

 

RPGLE : WRKDATAC               

  FICHIER SOURCE  . . . . .  GBTEST/QRPGSRC 

  MEMBRE  . . . . . . . . .  INRRPG50 

    200       *   INRRPG50 : Mise en couleur des Sources RPG                                                         

   1000       *                                                                                                          

   1100       *-----------------------------------------------------------------------*                                  

   1200      FINPRPGSRUP  E                    DISK                                                                      

   1300      E                    OP        100  5               Codes Opérations                                        

   1400      E                    SQ        100  4               N° Ligne Source                                         

   1500       *-----------------------------------------------------------------------*                                  

   1600      IDATA        DS                                                                                             

   1700      I                                        1   4 LIG                                                          

   1800      I                                        1   3 TABLE                                                        

   1900      I                                        5   5 BIT                                                          

   2000      I                                        6   6 CARTE                                                        

   2100      I                                        7   7 COMM                                                         

   2200      I                                        7  14 FICH                                                         

   2300      I                                       28  32 OPER                                                         

   2400      I                                       28  29 OPIF                                                         

   2500      I                                       28  30 OPCAS                                                        

   2600      I                                       28  30 OPEND                                                        

   2700      I                                       31  32 OPENDZ                                                       

   2800      I                                       28  31 OPELSE                                                       

   2900      I                                       54  55 INHI                                                         

   3000      I                                       56  57 INLO                                                         

   3100      I                                       58  59 INEQ                                                         

   3200      I                                       69  80 WFIN                                                         

   3300      I                                       69  70 WFINDO                                                       

   3400      I                                       71  76 WFINOP                                                       

   3500      I                                       77  80 WFINSQ                                                       

   3600      I                                       75  79 INBLAN                                                       

   3700      I                                       80  80 POS80                                                        

   3800      IINDIC       DS                                                                                             

   3900      I                                        1   1 HI                                                           

   4000      I                                        3   3 LO                                                           

   4100      I                                        5   5 EQ                                                           

   4200      I                                        6   6 BL                                                           

   4300       *-----------------------------------------------------------------------*                                  

   4400      C           *ENTRY    PLIST                                                                                 

   4500      C                     PARM           NBRA    4                                                              

   4600       *-----------------------------------------------------------------------*                                  

   4700      C  N10                EXSR $INIT                                                                            

   4800      C                     MOVELSRCDTA    DATA                                                                   

   4900      C                     CLEARINDIC                                                                            

   5000      C                     MOVE ' '       BITD    1                                                              

   5100      C                     MOVEL'    '    INCR    4                                                              

   5200 D001 C           CPT       IFNE 000                                                                              

   5300  001 C                     MOVE CPT       INCR    4                                                              

   5400 F001 C                     END                                      < IFNE  0052                                 

   5500      C           TABLE     CABEQ'** '     £FIN           99  Table compil.    =                                  

   5600       * Commentaires                                                                                             

   5700 D001 C           COMM      IFEQ '*'                                                                              

   5800  001 C                     MOVEL'    '    INCR    4                                                              

   5900  001 C                     MOVE BIT22     BITD    1        * Blanc                                               

   6100  001 C                     GOTO £LIGF                                                                            

   6200 F001 C                     END                                      < IFEQ  0057                                 

   6300       * Test Carte Gap                                                                                           

   6400 D001 C           CARTE     IFNE 'H'                                                                              

   6500  001 C           CARTE     ANDNE'F'                                                                              

   6600  001 C           CARTE     ANDNE'E'                                                                              

   6700  001 C           CARTE     ANDNE'L'                                                                              

   6800  001 C           CARTE     ANDNE'I'                                                                              

   6900  001 C           CARTE     ANDNE'C'                                                                              

   7000  001 C           CARTE     ANDNE'O'                                                                              

   7100  001 C                     GOTO £FIN                                                                             

   7200 F001 C                     END                                      < IFNE  0064                                 

   7300       *-----------------------------------------------------------------------*                                  

   7900       * Fichier < F >                                                                                            

   8000 D001 C           CARTE     IFEQ 'F'                                                                              

   8100  001 C           FICH      ANDNE'       '                                                                        

   8200  001 C                     MOVE BIT34     BITD             * Turquoise,Sép,Soul.                                 

   8400 F001 C                     END                                      < IFEQ  0080                                 

   8500       *-----------------------------------------------------------------------*                                  

   8600       * Fichier < I >                                                                                            

   8700 D001 C           CARTE     IFEQ 'I'                                                                              

   8800  001 C           FICH      ANDNE'       '                                                                        

   8900  001 C                     MOVE BIT38     BITD             * Rose                                                

   9000 F001 C                     END                                      < IFEQ  0087                                 

   9100       *-----------------------------------------------------------------------*                                  

   9200       * Carte < C > Incréments de Boucle                                                                         

   9300 D001 C           CARTE     IFEQ 'C'                                                                              

   9400 D002 C           OPIF      IFEQ 'DO'                       DO...                                                 

   9500  002 C           OPIF      OREQ 'IF'                       IF..                                                  

   9600  002 C           OPER      OREQ 'SELEC'                    SELEC                                                 

   9700  002 C                     ADD  001       CPT                                                                    

   9800  002 C                     MOVEL'D'       INCR             Début                                                 

   9900  002 C                     MOVE CPT       INCR                                                                   

  10000 D003 C           CPT       IFGT 0                                                                                

  10100  003 C                     MOVELOPER      OP,CPT           Mémorise Opération                                    

  10200  003 C                     MOVELSRCSEQ    SQ,CPT              "     N° ligne                                     

  10300 F003 C                     END                                      < IFGT  0100                                 

  10400 F002 C                     END                                      < IFEQ  0094                                 

  10500 D002 C           OPCAS     IFEQ 'CAS'                                                                            

  10600  002 C                     MOVEL'CAS '    INCR                                                                   

  10700  002 C                     MOVE '*'       BITCAS  1                                                              

  10800 F002 C                     END                                      < IFEQ  0105                                 

  10900 D002 C           OPELSE    IFEQ 'ELSE'                                                                           

  11000  002 C                     MOVEL'X'       INCR                                                                   

  11100  002 C                     MOVE CPT       INCR                                                                   

  11200 F002 C                     END                                      < IFEQ  0109                                 

  11300 D002 C           OPEND     IFEQ 'END'                                                                            

  11400  002 C           OPENDZ    ANDNE'SR'                                                                             

  11500 D003 C           BITCAS    IFEQ '*'                                                                              

  11600  003 C                     MOVEL'CAS '    INCR                                                                   

  11700  003 C                     MOVE ' '       BITCAS                                                                 

  11800 X003 C                     ELSE                                                                                  

  11900  003 C                     MOVEL'F'       INCR             Fin IF ou DO OU SELEC                                 

  12000  003 C                     MOVE CPT       INCR                                                                   

  12100 D004 C           WFIN      IFEQ *BLANKS                    Mise en place                                         

  12200  004 C           WFINDO    OREQ '< '                         < IF... DO...                                       

  12300  004 C                     MOVEL'< '      WFINDO                    <                                            

  12400 D005 C           CPT       IFGT 0                                                                                

  12500  005 C                     MOVELOP,CPT    WFINOP                      Opcde                                      

  12600  005 C                     MOVE SQ,CPT    WFINSQ                            1254                                 

  12700 F005 C                     END                                      < IFGT  0124                                 

  12800  004 C                     MOVE WFIN      SRCDTA                    < DOUEQ 0001                                 

  12900 F004 C                     END                                      < IFEQ  0121                                 

  13000 D004 C           CPT       IFGT 0                                                                                

  13100  004 C                     MOVE *BLANKS   OP,CPT                                                                 

  13200 F004 C                     END                                      < IFGT  0130                                 

  13300  003 C                     SUB  001       CPT                                                                    

  13400 F003 C                     END                                      < IFEQ  0115                                 

  13500 F002 C                     END                                      < IFEQ  0113                                 

  13600       * Carte < C > INDICATEURS                                                                                  

  13700 D002 C           CARTE     IFEQ 'C'                                                                              

  13800  002 C           INBLAN    ANDEQ*BLANKS                                                                          

  13900 D003 C           INHI      IFNE '  '                                                                             

  14000  003 C                     MOVEL'>'       HI                                                                     

  14100 F003 C                     END                                      < IFNE  0139                                 

  14200 D003 C           INLO      IFNE '  '                                                                             

  14300  003 C                     MOVEL'<'       LO                                                                     

  14400 F003 C                     END                                      < IFNE  0142                                 

  14500 D003 C           INEQ      IFNE '  '                                                                             

  14600  003 C                     MOVEL'='       EQ                                                                     

  14700 F003 C                     END                                      < IFNE  0145                                 

  14800 D003 C           INDIC     IFNE *BLANKS                                                                          

  14900  003 C                     MOVE INDIC     SRCDTA                                                                 

  15000 F003 C                     END                                      < IFNE  0148                                 

  15100 F002 C                     END                                      < IFEQ  0137                                 

  15200 F001 C                     END                                      < IFEQ  0093                                 

  15300       * Carte < C > EXSR                                                                                         

  15400 D001 C           CARTE     IFEQ 'C'                                                                              

  15500  001 C           OPER      ANDEQ'EXSR '                    * Bleu                                                

  15600  001 C                     MOVE BIT3A     BITD                                                                   

  15700 F001 C                     END                                      < IFEQ  0154                                 

  15800       * Carte < C > BEGSR,ENDSR                                                                                  

  15900 D001 C           CARTE     IFEQ 'C'                                                                              

  16000  001 C           OPER      ANDEQ'BEGSR'                                                                          

  16100  001 C           OPER      OREQ 'ENDSR'                                                                          

  16200  001 C                     MOVE BIT28     BITD             * Rouge                                               

  16400 F001 C                     END                                      < IFEQ  0159                                 

  16500       * Carte < C > WRITE, UPDAT, DELET, EXFMT, EXCPT                                                            

  16600 D001 C           CARTE     IFEQ 'C'                                                                              

  16700  001 C           OPER      ANDEQ'WRITE'                                                                          

  16800  001 C           OPER      OREQ 'UPDAT'                                                                          

  16900  001 C           OPER      OREQ 'DELET'                                                                          

  17000  001 C           OPER      OREQ 'EXFMT'                                                                          

  17100  001 C           OPER      OREQ 'EXCPT'                                                                          

  17200  001 C           OPER      OREQ 'UNLCK'                                                                          

  17300  001 C                     MOVE BIT31     BITD             * Turquoise                                           

  17500 F001 C                     END                                      < IFEQ  0166                                 

  17600       * Carte < C > CHAIN, SETLL, SETGT, READ                                                                    

  17700 D001 C           CARTE     IFEQ 'C'                                                                              

  17800  001 C           OPER      ANDEQ'CHAIN'                                                                          

  17900  001 C           OPER      OREQ 'SETLL'                                                                          

  18000  001 C           OPER      OREQ 'SETGT'                                                                          

  18100  001 C           OPER      OREQ 'READ '                                                                          

  18200  001 C           OPER      OREQ 'READE'                                                                          

  18300  001 C           OPER      OREQ 'READP'                                                                          

  18400  001 C           OPER      OREQ 'READC'                                                                          

  18500  001 C                     MOVE BIT33     BITD             * Jaune inv.                                          

  18700 F001 C                     END                                      < IFEQ  0177                                 

  18800       * Carte < C > CALL                                                                                         

  18900 D001 C           CARTE     IFEQ 'C'                                                                              

  19000  001 C           OPER      ANDEQ'CALL '                                                                          

  19100  001 C           OPER      OREQ 'PLIST'                                                                          

  19200  001 C                     MOVE BIT39     BITD             * Rose inv.                                           

  19400 F001 C                     END                                      < IFEQ  0189                                 

  19500       *                                                                                                          

  19600 D001 C           CARTE     IFEQ 'C'                                                                              

  19700  001 C           OPIF      ANDEQ'IF'                                                                             

  19800  001 C                     MOVE BIT30     BITD                                                                   

  20000 F001 C                     END                                      < IFEQ  0196                                 

  20100 D001 C           CARTE     IFEQ 'C'                                                                              

  20200  001 C           OPIF      ANDEQ'DO'                                                                             

  20400  001 C                     MOVE BIT3A     BITD                                                                   

  20600 F001 C                     END                                      < IFEQ  0201                                 

  20700 D001 C           CARTE     IFEQ 'C'                                                                              

  20800  001 C           OPER      ANDEQ'KLIST'                                                                          

  20900  001 C                     MOVE BIT3A     BITD                                                                   

  21000 F001 C                     END                                      < IFEQ  0207                                 

  21200 D001 C           CARTE     IFEQ 'C'                                                                              

  21300  001 C           OPER      ANDEQ'ELSE'                                                                           

  21400  001 C                     MOVE BIT30     BITD                                                                   

  21500 F001 C                     END                                      < IFEQ  0212                                 

  21700       *-----------------------------------------------------------------------*                                  

  21800       * Fichier < O >                                                                                            

  21900 D001 C           CARTE     IFEQ 'O'                                                                              

  22000  001 C           FICH      ANDNE'       '                                                                        

  22100  001 C                     MOVE BIT36     BITD                                                                   

  22200 F001 C                     END                                      < IFEQ  0219                                 

  22300       * Sortie enreg.                                                                                            

  22400      C           £LIGF     TAG                                                                                   

  22500 D001 C           POS80     IFEQ BIT20                                                                            

  22600  001 C                     MOVE ' '       SRCDTA                                                                 

  22700 F001 C                     END                                      < IFEQ  0225                                 

  22800      C                     MOVELINCR      Z5      5                                                              

  22900      C                     MOVE BITD      Z5                                                                     

  23000      C                     MOVELZ5        SRCDTA                                                                 

  23200       *                                                                                                          

  23300      C                     UPDATRPGSRC                                                                           

  23400       *-----------------------------------------------------------------------*                                  

  23500      C           £FIN      TAG                                                                                   

  23600      C                     ADD  0001      NBR                                                                    

  23700      C                     MOVE NBR       NBRA                                                                   

  23800      C   99                RETRN                                                                                 

  23900       *-----------------------------------------------------------------------*                                  

  24000       *--------*                                                                                                 

  24100       * $INIT  :                                                                                                 

  24200       *--------*                                                                                                 

  24300      C           $INIT     BEGSR                                                                                 

  24400      C                     Z-ADD000       CPT     30                                                             

  24500      C                     Z-ADD0000      NBR     40                                                             

  24600      C                     BITOF'01234567'BIT20   1                                                              

  24700      C                     MOVE BIT20     BIT22   1                                                              

  24800      C                     MOVE BIT20     BIT28   1                                                              

  24900      C                     MOVE BIT20     BIT30   1                                                              

  25000      C                     MOVE BIT20     BIT31   1                                                              

  25100      C                     MOVE BIT20     BIT33   1                                                              

  25200      C                     MOVE BIT20     BIT34   1                                                              

  25300      C                     MOVE BIT20     BIT36   1                                                              

  25400      C                     MOVE BIT20     BIT38   1                                                              

  25500      C                     MOVE BIT20     BIT39   1                                                              

  25600      C                     MOVE BIT20     BIT3A   1                                                              

  25700      C                     BITON'2'       BIT20            * Normal                                              

  25800      C                     BITON'26'      BIT22            * X 2                                                 

  25900      C                     BITON'24'      BIT28            * Rouge                                               

  26000      C                     BITON'23'      BIT30                                                                  

  26100      C                     BITON'237'     BIT31            * Turquoise, inv.                                     

  26200      C                     BITON'236'     BIT33            * Jaune, inv.                                         

  26300      C                     BITON'235'     BIT34            * Sép.Soul,Turquoise                                  

  26400      C                     BITON'2356'    BIT36            * Jaune Sép.soul X 2                                  

  26500      C                     BITON'234'     BIT38            * Rose (Sép, Clign)                                   

  26600      C                     BITON'2347'    BIT39            * Rose Inv.                                           

  26700      C                     BITON'2346'    BIT3A            * Bleu                                                

  26800       *                                                                                                          

  26900      C                     SETON                     10                   >                                      

  27000      C                     ENDSR                                                                                 

 


